
กราฟกน้ําหนักบรรทุกบนคาน ตอนที่ 1 
ผศ. สรกานต  ศรตีองออน 

บทความตอนนี้วาดวยการสรางกราฟกน้ําหนักบรรทุกแผสม่ําเสมอ  (uniform  load)  เต็ม 
ชวงคาน ซ่ึงการสรางจะตอเนื่องจากบทความตอนที่แลว (การปรับเสกลหนาจอแบบพอดกีับกราฟก 
คาน) คือ เซ็ตพิกัดตามนั้นและพล็อตกราฟกคานขึน้มาแลว 

แนวคิด 
กรณนี้ําหนักบรรทุกแผสม่ําเสมอเต็มชวงคานนี้ 

Ø  พิกัดแกน X  จะเปนระยะจริง คือ  หนวยความยาวตามชวงคาน  ซ่ึงเปนหนวยที่เซ็ตสเกลไว 
ตามปกติ (ในตัวอยางนีค้ือหนวยm) 

Ø  พิกัดแกน Y เปนคาน้ําหนักบรรทุก (ในตัวอยางนีค้ือ T/m) จึงตองมีการแปลงสเกลใหอยูใน 
หนวยที่เซ็ตสเกลไว  โดยการเทียบคาน้าํหนักบรรทุกสูงสุดกับคาสูงสุดของสเกลท่ีเรา 
กําหนดขึ้น ซ่ึงมีวิธีการดังนี้ 
1.  กําหนดคาสูงสุดของสเกล (ใชชื่อตวัแปรMaxS ในที่นีก้ําหนดคาเทากับ 1 m) 
2.  หาคาสูงสุดของน้ําหนักบรรทุกตามขอมูลที่ปอนเขาไป (Wmax) 
3.  พิกัดแกน  Y  ของน้ําหนกับรรทกุใดๆ  หาคาไดจากการเทียบน้ําหนักบรรทุกใดๆ  (W) 

กับคาสูงสุด ดังนี้ 

โดยที่  พิกัดแกน Y ของWmax  =  MaxS 
ดังนั้น  พิกดัแกน Y ของW  =  W * MaxS / Wmax 

=  kw * W 
เม่ือ kw = MaxS / Wmax 

Ø  เม่ือไดคา  kw*W  ซ่ึงเปนหนวยตามสเกลในที่นีค้ือหนวย  m  แลว  ก็นําไปรวมกับพกิัดแกน 
Y  ซ่ึงอางอิงจากพิกัดแนวแผนภาพวตัถุอิสระ  (FBD)  ของคาน  ก็จะไดพิกดัแกน  Y  ของ 
น้ําหนักบรรทุก เพ่ือนําไปพล็อต

Ø  การสรางกราฟกน้ําหนักบรรทุกแผแตละชวงคาน จะเปนรปูส่ีเหล่ียม มีพิกัดมุมบนซายและ 
มุมลางขวา ดังรูปที่ 1 

รูปที่ 1  พิกัดของกราฟกน้ําหนักบรรทุกแผสม่ําเสมอ 

ซ่ึงใชคําส่ัง Line (X, Yw)-(X + L(I), Yr), QBColor(คาของสีที่ใช), B 
โดยที ่ X  =  ระยะวัดจากปลายซายของคานชวงที่พล็อต 

Yr  =  พิกัดแกน Y อางอิงจากแนว FBD ของคาน 
Yw =  Yr + (kW * W) 

การโปรแกรม 
โปรแกรมจะตอเนื่องจากบทความตอนที่แลว โดยส่ิงที่เพิ่มเติมเขามาคือ 
1.  ตัวแปรเกี่ยวกับน้ําหนักบรรทกุ 
2.  ขอมูลที่สมมุติขึ้นสําหรับน้ําหนกับรรทกุ (เพื่อแสดงการสรางกราฟก) 
3.  โปรแกรมยอยสําหรับสรางกราฟกน้ําหนักบรรทุกของคานชื่อ Sub PlotUniformLoad 

ซ่ึงเขียนขึ้นตามแนวคิดที่ไดอธิบายในบทความนี้ 

เพ่ือใหเห็นภาพรวมของโปรแกรมทั้งหมด  จึงขอยกสวนของบทความตอนที่แลวมารวม 
ดวย โดยมีรหัสโปรแกรมทั้งหมดดังนี้ 

--------------------------------------------------------------------------------------------------------------------- 
Option Explicit 
Dim Xmin As Single: Dim Xmax As Single 
Dim Ymin As Single: Dim Ymax As Single 
Dim StepX As Single: Dim StepY As Single 
Const N = 3 ' สมมุติจํานวนชวงคานเทากับ 3

Dim L(0 To N + 1) As Single ' ความยาวแตละชวงคาน 
Dim SumL As Single ' ผลรวมความยาวของแตละชวงคาน 
' รูปแบบคาน : 1 = ไมมีชวงยื่น, 2 = ปลายยื่นซาย 
' 3 = ปลายยื่นขวา, 4 = ปลายยื่นสองดาน 
Dim BeamType As Integer 
Dim S As Integer ' คาเริ่มตนของลูป 
Dim E As Integer ' คาจบของลูป 
Dim w(0 To N + 1) As Single ' น้ําหนักบรรทุกแผเต็มชวง แตละชวงคาน 
--------------------------------------------------------------------------------------------------------------------- 
Sub PlotUniformLoad() 
Const MaxS = 1 
Dim kw As Single 
Dim Wmax As Single 
Dim I As Integer 
Dim X As Single 
Dim Yw As Single 
Dim Yr As Single 
Dim strW As String 

' หาคาน้ําหนกับรรทกุสูงสุด 
Wmax = w(S) 
For I = S + 1 To E 

If Wmax < w(I) Then 
Wmax = w(I) 

End If 
Next I 
' ถาไมมี Uniform Load 
If Wmax = 0 Then 

Exit Sub 
End If

' แปลงสเกล 
kw = MaxS / Wmax 
' พล็อต 
DrawWidth = 1 
X = 0: Yr = Ymax - 5 + 0.1 
For I = S To E 

If w(I) <> 0 Then 
Yw = Yr + kw * w(I) 
Line (X, Yw)-(X + L(I), Yr), QBColor(12), B 
strW = Format(w(I), "0.000") 
CurrentX = X + ((L(I) - TextWidth(strW)) / 2) 
CurrentY = Yw - TextHeight("M") 
Print strW 

End If 
X = X + L(I) 

Next I 
End Sub 
--------------------------------------------------------------------------------------------------------------------- 
Sub PlotBeam() 
Dim I As Integer 
Dim X As Single: Dim Y As Single 
Dim d As String 

' สรางกราฟกคานแตละชวง 
X = 0: Y = Ymax – 5 
For I = S To E 

Line (X, Y + 0.1)-(X + L(I), Y), QBColor(1), B 
X = X + L(I) 

Next I 
' เสนบอกความยาวแตละชวง

X = 0: Y = Ymax – 7 
Line (X, Y)-(X + SumL, Y), QBColor(4) 
Line (X, Y + 0.5)-(X, Y - 0.5), QBColor(4) 
For I = S To E 

X = X + L(I) 
Line (X, Y + 0.5)-(X, Y - 0.5), QBColor(4) 

Next I 
' แสดงคาความยาวแตละชวง 
X = 0: Y = Ymax – 7 
For I = S To E 

d = Format(L(I), "0.00") 
CurrentX = X + (L(I) / 2) - (TextWidth(d) / 2) 
CurrentY = Y - TextHeight(d) 
Print d 
X = X + L(I) 

Next I 
' เสนแรงปฏิกิริยา 
X = 0: Y = Ymax – 5 
Select Case BeamType 

Case 2, 4 ' ปลายยื่นซาย, ปลายยื่นสองดาน 
X = X + L(0) 

End Select 
DrawWidth = 3 
Line (X, Y)-(X, Y - 1), QBColor(1) 
Line (X, Y)-(X - 0.1, Y - 0.1), QBColor(1) 
Line (X, Y)-(X + 0.1, Y - 0.1), QBColor(1) 
DrawWidth = 1 
For I = 1 To N 

X = X + L(I) 
DrawWidth = 3

Line (X, Y)-(X, Y - 1), QBColor(1) 
Line (X, Y)-(X - 0.1, Y - 0.1), QBColor(1) 
Line (X, Y)-(X + 0.1, Y - 0.1), QBColor(1) 
DrawWidth = 1 

Next I 
End Sub 
--------------------------------------------------------------------------------------------------------------------- 
Sub FitScreen() 
Dim I As Integer 

' หาคาเริ่มตนและคาจบของลูป 
Select Case BeamType 

Case 1 ' ไมมีชวงยื่น 
S = 1: E = N 

Case 2 ' ปลายยื่นซาย 
S = 0: E = N 

Case 3 ' ปลายยื่นขวา 
S = 1: E = N + 1 

Case 4 ' ปลายยื่นสองดาน 
S = 0: E = N + 1 

End Select 
' หาผลรวมความยาวของแตละชวงคาน 
SumL = 0 
For I = S To E 

SumL = SumL + L(I) 
Next I 
' กําหนดสเกลหนาจอแกน X เซ็ตขอบซายที่ -2 m เสมอ 
Xmin = -2: Xmax = SumL + 2 
' กําหนดสเกลหนาจอแกน Y เทากับ 3/4 เทาของแกน X โดยเซ็ตขอบลางท่ี -2 m เสมอ 
Ymin = -2: Ymax = (3 * (Xmax + Abs(Xmin)) \ 4) - Abs(Ymin)

End Sub 
--------------------------------------------------------------------------------------------------------------------- 
Private  Sub  Form_MouseMove(Button  As  Integer,  Shift  As  Integer,  X  As  Single,  Y  As 
Single) 

' แสดงตําแหนงพิกัดที่ StatusBar ทศนิยมหนึ่งตําแหนง 
stbMain.SimpleText = "พิกัดแกน X = " & Format(X, "0.0") 

End Sub 
--------------------------------------------------------------------------------------------------------------------- 
Private Sub Form_Activate() 
Dim Row As Integer: Dim Col As Integer 
Dim NumX As Single: Dim NumY As Single 
Dim X As Single: Dim Y As Single 

Me.Scale (Xmin, Ymax)-(Xmax, Ymin) ' กําหนดสเกลการแสดงผล 
NumX = (Abs(Xmax - Xmin) / StepX) + 1 ' หาจํานวนจดุกริดในแกน X 
NumY = (Abs(Ymax - Ymin) / StepY) + 1 ' หาจํานวนจดุกริดในแกน Y 
Cls ' เคลียรหนาจอ 
' พล็อตจุดกรดิ 
X = Xmin: Y = Ymin 
For Row = 1 To NumX 

For Col = 1 To NumY 
PSet (X, Y), QBColor(4) 
Y = Y + StepY 

Next Col 
X = X + StepX: Y = Ymin 

Next Row 
' ลากเสนแกนอางอิงพิกัด (0,0) 
Line (0, 0)-(1.5, 0), QBColor(4): Line (0, 0)-(0, 1.5), QBColor(4) 
' เรียกโปรแกรมยอยสําหรับสรางกราฟกคาน 
PlotBeam

' เรียกโปรแกรมยอยสําหรับสรางกราฟกน้ําหนักบรรทุกแผสม่ําเสมอเต็มชวง 
PlotUniformLoad 

End Sub 
--------------------------------------------------------------------------------------------------------------------- 
Private Sub Form_Load() 
Dim I As Integer 

Me.WindowState = 2 ' ขยายฟอรมใหเต็มจอภาพ 
Me.ScaleMode = 0 ' กําหนดสเกลหนาจอเอง 
Me.BackColor = vbWhite ' สีฉากหลังเปนสีขาว 
Me.AutoRedraw = True ' จัดเก็บการแสดงผลในหนวยความจํา 

' ******************** ขอมูลที่สมมุติขึ้น ************************ 
' สมมุติคานปลายยื่นสองดาน 
BeamType = 4 
' ความยาวคาน หนวยm 
L(0) = 2: L(1) = 6: L(2) = 5: L(N) = 8: L(N + 1) = 1.5 
' น้ําหนักบรรทุก หนวย T/m 
w(0) = 1.234: w(1) = 0: w(2) = 3.456: w(N) = 2.345: w(N + 1) = 0.98 
' ************************************************************* 
FitScreen 
StepX = 1: StepY = 1 ' กําหนดระยะในการแสดงจุดกรดิ 
Call Form_Activate ' ไปท่ีเหตุการณ Activate 

End Sub 
--------------------------------------------------------------------------------------------------------------------- 

เม่ือทําการแสดงผล จะเห็นกราฟกน้ําหนกับรรทกุแผ และ FBD ของคาน ดังรูปที่ 2

รูปที่ 2  กราฟกที่ไดเม่ือทําการแสดงผล 

